
Euler menyediakan beberapa fungsi untuk melakukan visualisasi dan perhitungan geometri, baik secara
numerik maupun analitik (seperti biasanya tentunya, menggunakan Maxima). Fungsi-fungsi untuk
visualisasi dan perhitungan geometeri tersebut disimpan di dalam file program "geometry.e", sehingga file
tersebut harus dipanggil sebelum menggunakan fungsi-fungsi atau perintah-perintah untuk geometri.

>load geometry

Numerical and symbolic geometry.

Fungsi-fungsi Geometri

Fungsi-fungsi untuk Menggambar Objek Geometri:

  defaultd:=textheight()*1.5: nilai asli untuk parameter d
  setPlotrange(x1,x2,y1,y2): menentukan rentang x dan y pada bidang

koordinat
setPlotRange(r): pusat bidang koordinat (0,0) dan batas-batas sumbu-x dan y adalah -r sd r
plotPoint (P, "P"): menggambar titik P dan diberi label "P"
plotSegment (A,B, "AB", d): menggambar ruas garis AB, diberi label "AB" sejauh d
plotLine (g, "g", d): menggambar garis g diberi label "g" sejauh d
plotCircle (c,"c",v,d): Menggambar lingkaran c dan diberi label "c"
plotLabel (label, P, V, d): menuliskan label pada posisi P

Fungsi-fungsi Geometri Analitik (numerik maupun simbolik):

  turn(v, phi): memutar vektor v sejauh phi
  turnLeft(v):   memutar vektor v ke kiri
  turnRight(v):  memutar vektor v ke kanan
  normalize(v): normal vektor v
  crossProduct(v, w): hasil kali silang vektorv dan w.
  lineThrough(A, B): garis melalui A dan B, hasilnya [a,b,c] sdh.

ax+by=c.
lineWithDirection(A,v): garis melalui A searah vektor v
getLineDirection(g): vektor arah (gradien) garis g
getNormal(g): vektor normal (tegak lurus) garis g
getPointOnLine(g): titik pada garis g
perpendicular(A, g): garis melalui A tegak lurus garis g
parallel (A, g): garis melalui A sejajar garis g
lineIntersection(g, h): titik potong garis g dan h
projectToLine(A, g): proyeksi titik A pada garis g
distance(A, B): jarak titik A dan B
distanceSquared(A, B): kuadrat jarak A dan B
quadrance(A, B): kuadrat jarak A dan B
areaTriangle(A, B, C): luas segitiga ABC
computeAngle(A, B, C): besar sudut <ABC
angleBisector(A, B, C): garis bagi sudut <ABC
circleWithCenter (A, r): lingkaran dengan pusat A dan jari-jari r
getCircleCenter(c): pusat lingkaran c
getCircleRadius(c): jari-jari lingkaran c
circleThrough(A,B,C): lingkaran melalui A, B, C
middlePerpendicular(A, B): titik tengah AB
lineCircleIntersections(g, c): titik potong garis g dan lingkran c
circleCircleIntersections (c1, c2): titik potong lingkaran c1 dan c2
planeThrough(A, B, C): bidang melalui titik A, B, C

Fungsi-fungsi Khusus Untuk Geometri Simbolik:

  getLineEquation (g,x,y): persamaan garis g dinyatakan dalam x dan y
  getHesseForm (g,x,y,A): bentuk Hesse garis g dinyatakan dalam x dan

y dengan titik A pada sisi positif (kanan/atas) garis
quad(A,B): kuadrat jarak AB
spread(a,b,c): Spread segitiga dengan panjang sisi-sisi a,b,c, yakni sin(alpha)^2 dengan alpha sudut yang
menghadap sisi a.
crosslaw(a,b,c,sa): persamaan 3 quads dan 1 spread pada segitiga dengan panjang sisi a, b, c.
triplespread(sa,sb,sc): persamaan 3 spread sa,sb,sc yang memebntuk suatu segitiga
doublespread(sa): Spread sudut rangkap Spread 2*phi, dengan sa=sin(phi)^2 spread a.

Contoh 1: Luas, Lingkaran Luar, Lingkaran Dalam Segitiga

Untuk menggambar objek-objek geometri, langkah pertama adalah menentukan rentang sumbu-sumbu
koordinat. Semua objek geometri akan digambar pada satu bidang koordinat, sampai didefinisikan bidang
koordinat yang baru.

>setPlotRange(-0.5,2.5,-0.5,2.5); // mendefinisikan bidang koordinat baru 

Sekarang atur tiga poin dan plot.

>A=[1,0]; plotPoint(A,"A"); // definisi dan gambar tiga titik
>B=[0,1]; plotPoint(B,"B");



>C=[2,2]; plotPoint(C,"C");

Lalu tiga segmen.

>plotSegment(A,B,"c"); // c=AB
>plotSegment(B,C,"a"); // a=BC
>plotSegment(A,C,"b"); // b=AC

Fungsi geometri meliputi fungsi untuk membuat garis dan lingkaran. Format untuk garis adalah [a, b, c],
yang merepresentasikan garis dengan persamaan ax + by = c.

>lineThrough(B,C) // garis yang melalui B dan C

[-1,  2,  2]

Hitung garis tegak lurus melalui A pada BC.

>h=perpendicular(A,lineThrough(B,C)); // garis h tegak lurus BC melalui A

Dan perpotongannya dengan BC.

>D=lineIntersection(h,lineThrough(B,C)); // D adalah titik potong h dan BC

Plot itu.

>plotPoint(D,value=1); // koordinat D ditampilkan
>aspect(1); plotSegment(A,D): // tampilkan semua gambar hasil plot...()

Hitung luas ABC:

>norm(A-D)*norm(B-C)/2 // AD=norm(A-D), BC=norm(B-C)

1.5

Cara lain menghitung rumus determinan.

>areaTriangle(A,B,C) // hitung luas segitiga langusng dengan fungsi

1.5

Cara lain menghitung luas segitigas ABC:

>distance(A,D)*distance(B,C)/2

1.5

Sudut di C

>degprint(computeAngle(B,C,A))

36°52'11.63''

Sekarang lingkaran sirkit segitiga.



>c=circleThrough(A,B,C); // lingkaran luar segitiga ABC
>R=getCircleRadius(c); // jari2 lingkaran luar 
>O=getCircleCenter(c); // titik pusat lingkaran c 
>plotPoint(O,"O"); // gambar titik "O"
>plotCircle(c,"Lingkaran luar segitiga ABC"):

Tampilkan koordinat titik pusat dan jari-jari lingkaran luar.

>O, R

[1.16667,  1.16667]
1.17851130198

Sekarang akan digambar lingkaran dalam segitiga ABC. Titik pusat lingkaran dalam adalah titik potong
garis-garis bagi sudut.

>l=angleBisector(A,C,B); // garis bagi <ACB
>g=angleBisector(C,A,B); // garis bagi <CAB
>P=lineIntersection(l,g) // titik potong kedua garis bagi sudut

[0.86038,  0.86038]

Tambahkan semua ke plot.

>color(5); plotLine(l); plotLine(g); color(1); // gambar kedua garis bagi sudut
>plotPoint(P,"P"); // gambar titik potongnya
>r=norm(P-projectToLine(P,lineThrough(A,B))) // jari-jari lingkaran dalam

0.509653732104

>plotCircle(circleWithCenter(P,r),"Lingkaran dalam segitiga ABC"): // gambar lingkaran dalam



Latihan

1. Tentukan ketiga titik singgung lingkaran dalam dengan sisi-sisi segitiga ABC.
2. Gambar segitiga dengan titik-titik sudut ketiga titik singgung tersebut.
3. Tunjukkan bahwa garis bagi sudut yang ke tiga juga melalui titik pusat lingkaran dalam.
4. Gambar jari-jari lingkaran dalam.

Jawab:
1. Tentukan ketiga titik singgung lingkaran dalam dengan sisi-sisi segitiga ABC.
Titik singgung garis BC dengan lingkaran dalam.

>s=lineThrough(B,C)

[-1,  2,  2]

>m=circleWithCenter(P,r)

[0.86038,  0.86038,  0.509654]

>S=lineCircleIntersections(s,m)

[0.632456,  1.31623]

Titik singgung garis AC dengan lingkaran dalam.

>p=lineThrough(A,C)

[-2,  1,  -2]

>Q=lineCircleIntersections(p,m)

[1.31623,  0.632456]

Titik singgung garis AB dengan lingkaran dalam.

> q=lineThrough(A,B)

[-1,  -1,  -1]

>L=lineCircleIntersections(q,m)

[0.5,  0.5]

Jadi titik singgung lingkaran dalam dengan sisi-sisi segitiga adalah (0.632456, 1.31623), (1.31632,
0.632456), (0.5, 0.5).

2. Gambar segitiga dengan titik-titik sudut ketiga titik singgung tersebut.

>plotSegment(S,Q,"a");
>plotSegment(S,L,"b");
>plotSegment(L,Q,"c"):

3. Tunjukkan bahwa garis bagi sudut yang ke tiga juga melalui titik pusat lingkaran dalam.



> P, r

[0.86038,  0.86038]
0.509653732104

>k=angleBisector(A,B,C)

[-0.264911,  -1.63246,  -1.63246]

>color(2); plotLine(k):

4. Gambar jari-jari lingkaran dalam.

>plotSegment(P,L,"r"):

Contoh 2: Geometri Smbolik
Kita dapat menghitung geometri tepat dan simbolis menggunakan Maxima.

Geometri file.e menyediakan fungsi yang sama (dan lebih banyak lagi) di Maxima. Namun, sekarang kita
dapat menggunakan perhitungan simbolik.

>A &= [1,0]; B &= [0,1]; C &= [2,2]; // menentukan tiga titik A, B, C

Fungsi garis dan lingkaran bekerja seperti fungsi Euler, tetapi menyediakan penghitungan simbolik.

>c &= lineThrough(B,C) // c=BC

                             [- 1, 2, 2]



Kita bisa mendapatkan persamaan untuk sebuah garis dengan mudah.

>$getLineEquation(c,x,y), $solve(%,y) | expand // persamaan garis c

>$getLineEquation(lineThrough(A,[x1,y1]),x,y) // persamaan garis melalui A dan (x1, y1)

>h &= perpendicular(A,lineThrough(B,C)) // h melalui A tegak lurus BC

                              [2, 1, 2]

>Q &= lineIntersection(c,h) // Q titik potong garis c=BC dan h

                                 2  6
                                [-, -]
                                 5  5

>$projectToLine(A,lineThrough(B,C)) // proyeksi A pada BC

>$distance(A,Q) // jarak AQ

>cc &= circleThrough(A,B,C); $cc // (titik pusat dan jari-jari) lingkaran melalui A, B, C

>r&=getCircleRadius(cc); $r , $float(r) // tampilkan nilai jari-jari

>$computeAngle(A,C,B) // nilai <ACB

>$solve(getLineEquation(angleBisector(A,C,B),x,y),y)[1] // persamaan garis bagi <ACB

>P &= lineIntersection(angleBisector(A,C,B),angleBisector(C,B,A)); $P // titik potong 2 garis bagi sudut

>P() // hasilnya sama dengan perhitungan sebelumnya

[0.86038,  0.86038]

Garis dan Lingkaran yang Berpotongan

Tentu saja, kita juga bisa memotong garis dengan lingkaran, dan lingkaran dengan lingkaran.

>A &:= [1,0]; c=circleWithCenter(A,4);
>B &:= [1,2]; C &:= [2,1]; l=lineThrough(B,C);
>setPlotRange(5); plotCircle(c); plotLine(l);

Perpotongan garis dengan lingkaran mengembalikan dua titik dan jumlah titik perpotongan.

>{P1,P2,f}=lineCircleIntersections(l,c);
>P1, P2,



[4.64575,  -1.64575]
[-0.645751,  3.64575]

>plotPoint(P1); plotPoint(P2):

Hal yang sama di Maxima.

>c &= circleWithCenter(A,4) // lingkaran dengan pusat A jari-jari 4

                              [1, 0, 4]

>l &= lineThrough(B,C) // garis l melalui B dan C

                              [1, 1, 3]

>$lineCircleIntersections(l,c) | radcan, // titik potong lingkaran c dan garis l

Akan ditunjukkan bahwa sudut-sudut yang menghadap bsuusr yang sama adalah sama besar.

>C=A+normalize([-2,-3])*4; plotPoint(C); plotSegment(P1,C); plotSegment(P2,C);
>degprint(computeAngle(P1,C,P2))

69°17'42.68''

>C=A+normalize([-4,-3])*4; plotPoint(C); plotSegment(P1,C); plotSegment(P2,C);
>degprint(computeAngle(P1,C,P2))

69°17'42.68''

>insimg;



Garis Sumbu
Berikut adalah langkah-langkah menggambar garis sumbu ruas garis AB:

1. Gambar lingkaran dengan pusat A melalui B.
2. Gambar lingkaran dengan pusat B melalui A.
3. Tarik garis melallui kedua titik potong kedua lingkaran tersebut. Garis ini merupakan garis sumbu
(melalui titik tengah dan tegak lurus) AB.

>A=[2,2]; B=[-1,-2];
>c1=circleWithCenter(A,distance(A,B));
>c2=circleWithCenter(B,distance(A,B));
>{P1,P2,f}=circleCircleIntersections(c1,c2);
>l=lineThrough(P1,P2);
>setPlotRange(5); plotCircle(c1); plotCircle(c2);
>plotPoint(A); plotPoint(B); plotSegment(A,B); plotLine(l):

Selanjutnya, kita melakukan hal yang sama di Maxima dengan koordinat umum.

>A &= [a1,a2]; B &= [b1,b2];
>c1 &= circleWithCenter(A,distance(A,B));
>c2 &= circleWithCenter(B,distance(A,B));
>P &= circleCircleIntersections(c1,c2); P1 &= P[1]; P2 &= P[2];

Persamaan untuk persimpangan cukup terlibat. Tapi kita bisa menyederhanakan, jika kita menyelesaikan
y.

>g &= getLineEquation(lineThrough(P1,P2),x,y);
>$solve(g,y)

Ini memang sama dengan tengah tegak lurus, yang dihitung dengan cara yang sama sekali berbeda.

>$solve(getLineEquation(middlePerpendicular(A,B),x,y),y)



>h &=getLineEquation(lineThrough(A,B),x,y);
>$solve(h,y)

Perhatikan hasil kali gradien garis g dan h adalah:

Artinya kedua garis tegak lurus.

Contoh 3: Rumus Heron
Rumus Heron menyatakan bahwa luas segitiga dengan panjang sisi-sisi a, b dan c adalah:

Untuk membuktikan hal ini kita misalkan C(0,0), B(a,0) dan A(x,y), b=AC, c=AB. Luas segitiga ABC adalah

Nilai y didapat dengan menyelesaikan sistem persamaan:

>setPlotRange(-1,10,-1,8); plotPoint([0,0], "C(0,0)"); plotPoint([5.5,0], "B(a,0)");  ...
  plotPoint([7.5,6], "A(x,y)");

>plotSegment([0,0],[5.5,0], "a",25); plotSegment([5.5,0],[7.5,6],"c",15); ...
 plotSegment([0,0],[7.5,6],"b",25);
>plotSegment([7.5,6],[7.5,0],"t=y",25):

>&assume(a>0); sol &= solve([x^2+y^2=b^2,(x-a)^2+y^2=c^2],[x,y])

                2    2    2
             - c  + b  + a
       [[x = --------------, y = 
                  2 a
          4      2  2      2  2    4      2  2    4
  sqrt(- c  + 2 b  c  + 2 a  c  - b  + 2 a  b  - a )
- --------------------------------------------------], 
                         2 a
        2    2    2
     - c  + b  + a
[x = --------------, y = 
          2 a
        4      2  2      2  2    4      2  2    4
sqrt(- c  + 2 b  c  + 2 a  c  - b  + 2 a  b  - a )
--------------------------------------------------]]
                       2 a

Ekstrak solusi y

>ysol &= y with sol[2][2]; $'y=sqrt(factor(ysol^2))



Kita mendapatkan formula Heron.

>function H(a,b,c) &= sqrt(factor((ysol*a/2)^2)); $'H(a,b,c)=H(a,b,c)

Tentu saja, setiap segitiga persegi panjang adalah kasus yang terkenal.

>H(3,4,5) //luas segitiga siku-siku dengan panjang sisi 3, 4, 5

6

Dan jelas juga, bahwa ini adalah segitiga dengan luas maksimal dan kedua sisinya 3 dan 4.

>aspect (1.5); plot2d(&H(3,4,x),1,7): // Kurva luas segitiga sengan panjang sisi 3, 4, x (1<= x <=7)

Kasus umum juga berfungsi.

>$solve(diff(H(a,b,c)^2,c)=0,c)

Sekarang mari kita cari himpunan semua titik di mana b + c = d untuk beberapa konstanta d. Diketahui
bahwa ini adalah elips.

>s1 &= subst(d-c,b,sol[2]); $s1

Dan membuat persamaan seperti ini

>function fx(a,c,d) &= rhs(s1[1]); $fx(a,c,d), function fy(a,c,d) &= rhs(s1[2]); $fy(a,c,d)

Sekarang kita bisa menggambar setnya. Sisi b bervariasi dari 1 hingga 4. Diketahui bahwa kita
mendapatkan elips.

>aspect(1); plot2d(&fx(3,x,5),&fy(3,x,5),xmin=1,xmax=4,square=1):



Kita dapat memeriksa persamaan umum elips ini, yaitu.

di mana (xm,ym) adalah pusat, dan u dan v adalah setengah sumbu.

>$ratsimp((fx(a,c,d)-a/2)^2/u^2+fy(a,c,d)^2/v^2 with [u=d/2,v=sqrt(d^2-a^2)/2])

Kita melihat bahwa tinggi dan luas segitiga adalah maksimal untuk x = 0. Jadi luas segitiga dengan a + b +
c = d adalah maksimal, jika sama sisi. Kami ingin mendapatkan ini secara analitis.

>eqns &= [diff(H(a,b,d-(a+b))^2,a)=0,diff(H(a,b,d-(a+b))^2,b)=0]; $eqns

Kami mendapatkan beberapa minima, yang termasuk dalam segitiga dengan satu sisi 0, dan solusi
a=b=c=d/3.

>$solve(eqns,[a,b])

Ada juga metode Lagrange, memaksimalkan H(a,b,c)^2 terhadap a+b+d=d.

>&solve([diff(H(a,b,c)^2,a)=la,diff(H(a,b,c)^2,b)=la, ...
    diff(H(a,b,c)^2,c)=la,a+b+c=d],[a,b,c,la])

                    d      d                d             d
       [[a = 0, b = -, c = -, la = 0], [a = -, b = 0, c = -, la = 0], 
                    2      2                2             2
                                                                  3
            d      d                       d      d      d       d
       [a = -, b = -, c = 0, la = 0], [a = -, b = -, c = -, la = ---]]
            2      2                       3      3      3       108

Kita bisa membuat plot situasinya.

Pertama, atur poin di Maxima

>A &= at([x,y],sol[2]); $A

>B &= [0,0]; $B, C &= [a,0]; $C

Kemudian atur rentang plot, dan plot poinnya.

>setPlotRange(0,5,-2,3); ...
 a=4; b=3; c=2; ...
 plotPoint(mxmeval("B"),"B"); plotPoint(mxmeval("C"),"C"); ...
 plotPoint(mxmeval("A"),"A"):



Variable b1 not found!
Use global variables or parameters for string evaluation.
Error in Evaluate, superfluous characters found.
Try "trace errors" to inspect local variables after errors.
mxmeval:
    return evaluate(mxm(s));
Error in:
... (0,5,-2,3); a=4; b=3; c=2; plotPoint(mxmeval("B"),"B"); plotPo ...
                                                     ^

Plot segmennya.

>plotSegment(mxmeval("A"),mxmeval("C")); ...
 plotSegment(mxmeval("B"),mxmeval("C")); ...
 plotSegment(mxmeval("B"),mxmeval("A")):

Variable a1 not found!
Use global variables or parameters for string evaluation.
Error in Evaluate, superfluous characters found.
Try "trace errors" to inspect local variables after errors.
mxmeval:
    return evaluate(mxm(s));
Error in:
plotSegment(mxmeval("A"),mxmeval("C")); plotSegment(mxmeval("B ...
                        ^

Hitung tengah tegak lurus di Maxima.

>h &= middlePerpendicular(A,B); g &= middlePerpendicular(B,C);

Dan bagian tengah dari keliling.

>U &= lineIntersection(h,g);

Kita mendapatkan rumus untuk jari-jari lingkaran.

>&assume(a>0,b>0,c>0); $distance(U,B) | radcan

Mari kita tambahkan ini ke plot.

>plotPoint(U()); ...
 plotCircle(circleWithCenter(mxmeval("U"),mxmeval("distance(U,C)"))):

Variable a2 not found!
Use global variables or parameters for string evaluation.
Error in expression: [-((a2-1)*b2^2+(-a2^2-a1^2+5)*b2+(a2-1)*b1^2+a2^2-5*a2+a1^2)/((2*a1-4)*b2+(2-2*a2)*b1
Error in:
plotPoint(U()); plotCircle(circleWithCenter(mxmeval("U"),mxmev ...
             ^

Menggunakan geometri, kita mendapatkan rumus sederhana

untuk radius. Kita dapat memeriksa, apakah ini benar dengan Maxima. Maxima akan menfaktorkannya
hanya jika kita mengkuadratkannya.

>$c^2/sin(computeAngle(A,B,C))^2  | factor

Contoh 4: Garis Euler dan Parabola
Garis euler adalah garis yang ditentukan dari segitiga yang tidak sama sisi. Ini adalah garis tengah
segitiga, dan melewati beberapa titik penting yang ditentukan dari segitiga, termasuk pusat ortosentrum,
sirkumenter, pusat massa, titik Exeter, dan pusat lingkaran sembilan titik segitiga.

Untuk demonstrasi, kami menghitung dan memplot garis Euler dalam segitiga.

Pertama, kami menentukan sudut segitiga di Euler. Kami menggunakan definisi, yang terlihat dalam
ekspresi simbolik.

>A::=[-1,-1]; B::=[2,0]; C::=[1,2];

Untuk memplot objek geometris, kami menyiapkan area plot, dan menambahkan poin ke dalamnya.
Semua plot objek geometris ditambahkan ke plot saat ini.

>setPlotRange(3); plotPoint(A,"A"); plotPoint(B,"B"); plotPoint(C,"C");

Kita juga bisa menambahkan sisi segitiga.

>plotSegment(A,B,""); plotSegment(B,C,""); plotSegment(C,A,""):



Berikut adalah luas segitiga menggunakan rumus determinan. Tentu saja kita harus mengambil nilai
absolut dari hasil ini.

>$areaTriangle(A,B,C)

Kita dapat menghitung koefisien dari sisi c.

>c &= lineThrough(A,B)

                            [- 1, 3, - 2]

Dan juga dapatkan rumus untuk baris ini.

>$getLineEquation(c,x,y)

Untuk bentuk Hesse, kita perlu menentukan titik, sehingga titik tersebut berada di sisi positif dari bentuk
Hesse. Memasukkan titik menghasilkan jarak positif ke garis.

>$getHesseForm(c,x,y,C), $at(%,[x=C[1],y=C[2]])

Sekarang kita menghitung sirkit ABC.

>LL &= circleThrough(A,B,C); $getCircleEquation(LL,x,y)

>O &= getCircleCenter(LL); $O

Plot lingkaran dan pusatnya. Cu dan U adalah simbolik. Kami mengevaluasi ekspresi ini untuk Euler.

>plotCircle(LL()); plotPoint(O(),"O"):



Kita dapat menghitung perpotongan ketinggian di ABC (orthocenter) secara numerik dengan perintah
berikut.

>H &= lineIntersection(perpendicular(A,lineThrough(C,B)),...
   perpendicular(B,lineThrough(A,C))); $H

Sekarang kita dapat menghitung garis Euler dari segitiga tersebut.

>el &= lineThrough(H,O); $getLineEquation(el,x,y)

Tambahkan ke plot kita.

>plotPoint(H(),"H"); plotLine(el(),"Garis Euler"):

Pusat gravitasi harus berada di garis ini.

>M &= (A+B+C)/3; $getLineEquation(el,x,y) with [x=M[1],y=M[2]]

>plotPoint(M(),"M"): // titik berat



Teorinya mengatakan bahwa MH=2*MO. Kita perlu menyederhanakan dengan radcan untuk mencapai ini.

>$distance(M,H)/distance(M,O)|radcan

Fungsinya termasuk fungsi untuk sudut juga.

>$computeAngle(A,C,B), degprint(%())

60°15'18.43''

Persamaan untuk pusat lingkaran tidak terlalu bagus.

>Q &= lineIntersection(angleBisector(A,C,B),angleBisector(C,B,A))|radcan; $Q

Mari kita hitung juga ekspresi jari-jari lingkaran yang tertulis.

>r &= distance(Q,projectToLine(Q,lineThrough(A,B)))|ratsimp; $r

>LD &=  circleWithCenter(Q,r); // Lingkaran dalam

Mari kita tambahkan ini ke plot.

>color(5); plotCircle(LD()):



Parabola
Selanjutnya akan dicari persamaan tempat kedudukan titik-titik yang berjarak sama ke titik C dan ke garis
AB.

>p &= getHesseForm(lineThrough(A,B),x,y,C)-distance([x,y],C); $p='0

Persamaan tersebut dapat digambar menjadi satu dengan gambar sebelumnya.

>plot2d(p,level=0,add=1,contourcolor=6):

Ini seharusnya menjadi beberapa fungsi, tetapi pemecah default Maxima dapat menemukan solusi hanya,
jika persamaan kita kuadratkan. Akibatnya, kami mendapatkan solusi palsu.

>akar &= solve(getHesseForm(lineThrough(A,B),x,y,C)^2-distance([x,y],C)^2,y)

        [y = - 3 x - sqrt(70) sqrt(9 - 2 x) + 26, 
                              y = - 3 x + sqrt(70) sqrt(9 - 2 x) + 26]

Solusi pertama adalah

maxima: akar[1]

Menambahkan solusi pertama ke pertunjukkan plot, bahwa itu memang jalan yang kita cari. Teori
mengatakan kepada kita bajwa itu adalah parabola yang diputar.

>plot2d(&rhs(akar[1]),add=1):



>function g(x) &= rhs(akar[1]); $'g(x)= g(x)// fungsi yang mendefinisikan kurva di atas

>T &=[-1, g(-1)]; // ambil sebarang titik pada kurva tersebut
>dTC &= distance(T,C); $fullratsimp(dTC), $float(%) // jarak T ke C

>U &= projectToLine(T,lineThrough(A,B)); $U // proyeksi T pada garis AB 

>dU2AB &= distance(T,U); $fullratsimp(dU2AB), $float(%) // jatak T ke AB

Ternyata jarak T ke C sama dengan jarak T ke AB. Coba Anda pilih titik T yang lain dan ulangi perhitungan-
perhitungan di atas untuk menunjukkan bahwa hasilnya juga sama.

Contoh 5: Trigonometri Rasional
Ini terinspirasi oleh ceramah N.J.Wildberger. Dalam bukunya "Proporsi Agung", Wildberger mengusulkan
untuk menggantikan pengertian klasik tentang jarak dan sudut dengan kuadransi dan penyebaran.
Dengan menggunakan ini, memang mungkin untuk menghindari fungsi trigonometri dalam banyak contoh,
dan tetap "rasional".

Berikut ini, saya memperkenalkan konsep, dan memecahkan beberapa masalah. Saya menggunakan
perhitungan simbolik Maxima di sini, yang menyembunyikan keuntungan utama dari trigonometri rasional
bahwa perhitungan dapat dilakukan dengan kertas dan pensil saja. Anda diundang untuk memeriksa hasil
tanpa komputer.

Intinya adalah bahwa perhitungan rasional simbolis sering kali menghasilkan hasil yang sederhana.
Sebaliknya, trigonometri klasik menghasilkan hasil trigonometri yang rumit, yang mengevaluasi ke
pendekatan numerik saja.

>load geometry;

Untuk pendahuluan pertama, kami menggunakan segitiga persegi panjang dengan proporsi Mesir terkenal
3, 4 dan 5. Perintah berikut adalah perintah Euler untuk memplot geometri bidang yang terdapat dalam file
Euler "geometry.e".

>C&:=[0,0]; A&:=[4,0]; B&:=[0,3]; ...
 setPlotRange(-1,5,-1,5); ...
 plotPoint(A,"A"); plotPoint(B,"B"); plotPoint(C,"C"); ...
 plotSegment(B,A,"c"); plotSegment(A,C,"b"); plotSegment(C,B,"a"); ...
 insimg(30);



Tentu saja,

di mana wa adalah sudut di A. Cara biasa untuk menghitung sudut ini, adalah dengan melakukan invers
dari fungsi sinus. Hasilnya adalah sudut yang tidak dapat dicerna, yang hanya dapat dicetak secara
perkiraan.

>wa := arcsin(3/5); degprint(wa)

36°52'11.63''

Trigonometri rasional mencoba menghindari hal ini.

Pengertian pertama dari trigonometri rasional adalah kuadran, yang menggantikan jarak. Faktanya, itu
hanyalah kuadrat jarak. Berikut ini, a, b, dan c menunjukkan kuadran sisi-sisinya.

Teorema Pythogoras menjadi a+b=c lalu.

>a &= 3^2; b &= 4^2; c &= 5^2; &a+b=c

                               25 = 25

Gagasan kedua dari trigonometri rasional adalah penyebarannya. Spread mengukur bukaan antar baris.
Ini adalah 0, jika garis sejajar, dan 1, jika garis persegi panjang. Ini adalah kuadrat dari sinus sudut antara
dua garis.

Penyebaran garis AB dan AC pada gambar di atas didefinisikan sebagai

di mana a dan c adalah kuadrat dari segitiga persegi panjang mana pun dengan satu sudut di A.

>sa &= a/c; $sa

Ini lebih mudah dihitung daripada sudut, tentu saja. Tetapi Anda kehilangan properti yang sudut dapat
ditambahkan dengan mudah.

Tentu saja, kita dapat mengubah nilai perkiraan sudut wa menjadi sprad, dan mencetaknya sebagai
pecahan.

>fracprint(sin(wa)^2)

9/25

Hukum cosinus dari trgonometri klasik diterjemahkan menjadi "hukum silang" berikut.

Di sini a, b, dan c adalah kuadran dari sisi-sisi segitiga, dan sa adalah sebaran di sudut A. Sisi a, seperti
biasa, berlawanan dengan sudut A.

Hukum ini diimplementasikan dalam file geometry.e yang kami muat ke Euler.

>$crosslaw(aa,bb,cc,saa)



Dalam kasus kami, kita mendapatkan

>$crosslaw(a,b,c,sa)

Mari kita gunakan crosslaw ini untuk mencari sebaran di A. Untuk melakukan ini, kita menghasilkan
crosslaw untuk kuadran a, b, dan c, dan menyelesaikannya untuk sebaran yang tidak diketahui sa.

Anda dapat melakukan ini dengan tangan dengan mudah, tetapi saya menggunakan Maxima. Tentu saja,
kami mendapatkan hasilnya, kami sudah mendapatkannya.

>$crosslaw(a,b,c,x), $solve(%,x)

Kami sudah tahu ini. Definisi penyebaran adalah kasus khusus dari hukum lintas hukum.

Kita juga bisa menyelesaikan ini untuk umum a, b, c. Hasilnya adalah rumus yang menghitung sebaran
sudut segitiga berdasarkan kuadran ketiga sisinya.

>$solve(crosslaw(aa,bb,cc,x),x)

Kita bisa membuat fungsi dari hasilnya. Fungsi seperti itu sudah ditentukan dalam file geometry.e Euler.

>$spread(a,b,c)

Sebagai contoh, kita bisa menggunakannya untuk menghitung sudut segitiga bersisi

Hasilnya rasional, yang tidak mudah didapat jika kita menggunakan trigonometri klasik.

>$spread(a,a,4*a/7)

Ini adalah sudut dalam derajat.

>degprint(arcsin(sqrt(6/7)))

67°47'32.44''

Contoh lain

Sekarang, mari kita coba contoh yang lebih canggih.

Kami mengatur tiga sudut segitiga sebagai berikut.

>A&:=[1,2]; B&:=[4,3]; C&:=[0,4]; ...
 setPlotRange(-1,5,1,7); ...
 plotPoint(A,"A"); plotPoint(B,"B"); plotPoint(C,"C"); ...
 plotSegment(B,A,"c"); plotSegment(A,C,"b"); plotSegment(C,B,"a"); ...
 insimg;



Menggunakan Pythogoras, mudah untuk menghitung jarak antara dua titik. Saya pertama kali
menggunakan jarak fungsi file Euler untuk geometri. Jarak fungsi menggunakan geometri klasik.

>$distance(A,B)

Euler juga memiliki fungsi kuadrans antara dua titik.

Dalam contoh berikut, karena c + b bukan a, segitiga tidak persegi panjang.

>c &= quad(A,B); $c, b &= quad(A,C); $b, a &= quad(B,C); $a,

Pertama, mari kita hitung sudut tradisional. Fungsi computeAngle menggunakan metode biasa
berdasarkan perkalian titik dari dua vektor. Hasilnya adalah beberapa pendekatan floating point.

>wb &= computeAngle(A,B,C); $wb, $(wb/pi*180)()

32.4711922908

Menggunakan pensil dan kertas, kita bisa melakukan hal yang sama dengan hukum silang. Kami
memasukkan kuadran a, b, dan c ke dalam hukum silang dan menyelesaikan untuk x.

>$crosslaw(a,b,c,x), $solve(%,x),

Artinya, fungsi penyebaran yang didefinisikan dalam "geometry.e".

>sb &= spread(b,a,c); $sb

Maxima mendapatkan hasil yang sama dengan menggunakan trigonometri biasa, jika kita
memaksakannya. Itu menyelesaikan istilah sin (arccos (...)) menjadi hasil pecahan. Kebanyakan siswa
tidak dapat melakukan ini.

>$sin(computeAngle(A,B,C))^2

Setelah kita mendapatkan sebaran di B, kita bisa menghitung tinggi ha di sisi a. Ingat bahwa

Menurut definisi.

>ha &= c*sb; $ha

Gambar berikut telah diproduksi dengan program geometri C.a.R., yang dapat menggambar kuadran dan
menyebar.

gambar : (20) Rational_Geometry_CaR.png

Menurut definisi, panjang ha adalah akar kuadrat dari kuadrannya.

>$sqrt(ha)

Sekarang kita bisa menghitung luas segitiga. Jangan lupa, bahwa kita berurusan dengan kuadran!

>$sqrt(ha)*sqrt(a)/2

Rumus determinan yang biasa menghasilkan hasil yang sama.

>$areaTriangle(B,A,C)



Formula Heron
Sekarang, mari kita selesaikan masalah ini secara umum!

>&remvalue(a,b,c,sb,ha);

Pertama-tama kita menghitung spread di B untuk segitiga dengan sisi a, b, dan c. Kemudian kami
menghitung luas area yang dikuadratkan ("kuadrea"?), Memfaktorkannya dengan Maxima, dan kami
mendapatkan rumus Heron yang terkenal.

Memang, ini sulit dilakukan dengan pensil dan kertas.

>$spread(b^2,c^2,a^2), $factor(%*c^2*a^2/4)

Aturan Triple Spread
Kerugian dari spread adalah bahwa mereka tidak lagi hanya menambahkan sudut serupa.

Namun, tiga sebaran segitiga memenuhi aturan "penyebaran rangkap tiga" berikut.

>&remvalue(sa,sb,sc); $triplespread(sa,sb,sc)

Aturan ini berlaku untuk tiga sudut yang bertambah menjadi 180°.

Sejak penyebaran

sama, aturan penyebaran tiga kali lipat juga benar, jika

Karena penyebaran sudut negatif adalah sama, aturan penyebaran tiga kali lipat juga berlaku, jika

Misalnya, kita dapat menghitung sebaran sudut 60°. Ini 3/4. Persamaan memiliki solusi kedua, di mana
semua spread adalah 0.

>$solve(triplespread(x,x,x),x)

Sebaran 90° jelaslah 1. Jika dua sudut dijumlahkan menjadi 90°, penyebarannya menyelesaikan
persamaan penyebaran rangkap tiga dengan a, b, 1. Dengan perhitungan berikut kita mendapatkan
a+b=1.

>$triplespread(x,y,1), $solve(%,x)

Karena penyebaran 180°-t sama dengan penyebaran t, rumus penyebaran rangkap tiga juga berlaku, jika
satu sudut adalah jumlah atau perbedaan dari dua sudut lainnya.

Jadi kita bisa menemukan sebaran sudut berlipat ganda. Perhatikan bahwa ada dua solusi lagi. Kami
menjadikan ini sebuah fungsi.

>$solve(triplespread(a,a,x),x), function doublespread(a) &= factor(rhs(%[1]))

                            - 4 (a - 1) a

Pembagi Sudut

Ini situasinya, kita sudah tahu.

>C&:=[0,0]; A&:=[4,0]; B&:=[0,3]; ...
 setPlotRange(-1,5,-1,5); ...
 plotPoint(A,"A"); plotPoint(B,"B"); plotPoint(C,"C"); ...
 plotSegment(B,A,"c"); plotSegment(A,C,"b"); plotSegment(C,B,"a"); ...
 insimg;



Mari kita hitung panjang bisektor sudut pada A. Tapi kita ingin menyelesaikannya untuk umum a, b, c.

>&remvalue(a,b,c);

Jadi pertama-tama kita menghitung sebaran sudut terbagi di A, menggunakan rumus sebaran rangkap
tiga.

Masalah dengan rumus ini muncul lagi. Ini memiliki dua solusi. Kami harus memilih yang benar. Solusi
lainnya mengacu pada sudut terbagi 180°-wa.

>$triplespread(x,x,a/(a+b)), $solve(%,x), sa2 &= rhs(%[1]); $sa2

Mari kita periksa persegi panjang Mesir.

>$sa2 with [a=3^2,b=4^2]

Kami dapat mencetak sudut di Euler, setelah mentransfer penyebaran ke radian.

>wa2 := arcsin(sqrt(1/10)); degprint(wa2)

18°26'5.82''

Titik P adalah perpotongan dari garis bagi sudut dengan sumbu y.

>P := [0,tan(wa2)*4]

[0,  1.33333]

>plotPoint(P,"P"); plotSegment(A,P):



Mari kita periksa sudut dalam contoh spesifik kita.

>computeAngle(C,A,P), computeAngle(P,A,B)

0.321750554397
0.321750554397

Sekarang kita menghitung panjang bisektor AP.

Kita menggunakan teorema sinus di segitiga APC. Teorema ini menyatakan bahwa

memegang di segitiga apa pun. Persegi itu, itu diterjemahkan ke dalam apa yang disebut "hukum
penyebaran"

dimana a, b, c menunjukkan qudrance.

Karena BPA sebaran adalah 1-sa2, kita dapatkan darinya bisa / 1 = b / (1-sa2) dan dapat menghitung bisa
(kuadran garis-garis).

>&factor(ratsimp(b/(1-sa2))); bisa &= %; $bisa

Mari kita periksa rumus ini untuk nilai Mesir kita.

>sqrt(mxmeval("at(bisa,[a=3^2,b=4^2])")), distance(A,P)

4.21637021356
4.21637021356

Kami juga dapat menghitung P menggunakan rumus spread.

>py&=factor(ratsimp(sa2*bisa)); $py

Nilainya sama dengan yang kita dapatkan dengan rumus trigonometri.

>sqrt(mxmeval("at(py,[a=3^2,b=4^2])"))

1.33333333333

Sudut Akord

Perhatikan situasi berikut.

>setPlotRange(1.2); ...
 color(1); plotCircle(circleWithCenter([0,0],1)); ...
 A:=[cos(1),sin(1)]; B:=[cos(2),sin(2)]; C:=[cos(6),sin(6)]; ...
 plotPoint(A,"A"); plotPoint(B,"B"); plotPoint(C,"C"); ...
 color(3); plotSegment(A,B,"c"); plotSegment(A,C,"b"); plotSegment(C,B,"a"); ...
 color(1); O:=[0,0];  plotPoint(O,"0"); ...



 plotSegment(A,O); plotSegment(B,O); plotSegment(C,O,"r"); ...
 insimg;

Kita bisa menggunakan Maxima untuk menyelesaikan rumus sebaran rangkap tiga untuk sudut di pusat O
untuk r. Jadi kita mendapatkan rumus untuk jari-jari kuadrat dari keliling dalam hal kuadrat sisi.

Kali ini, Maxima menghasilkan beberapa angka nol yang kompleks, yang kita abaikan.

>&remvalue(a,b,c,r); // hapus nilai-nilai sebelumnya untuk perhitungan baru
>rabc &= rhs(solve(triplespread(spread(b,r,r),spread(a,r,r),spread(c,r,r)),r)[4]); $rabc

Kita bisa menjadikannya sebagai fungsi Euler.

>function periradius(a,b,c) &= rabc;

Mari kita periksa hasilnya untuk poin A, B, C kita.

>a:=quadrance(B,C); b:=quadrance(A,C); c:=quadrance(A,B);

Radiusnya memang 1.

>periradius(a,b,c)

1

Faktanya, penyebaran CBA hanya bergantung pada b dan c. Ini adalah teorema sudut akord.

>$spread(b,a,c)*rabc | ratsimp

Sebenarnya sebarannya adalah b/(4r), dan kita melihat bahwa sudut akor b adalah setengah dari sudut
tengah.

>$doublespread(b/(4*r))-spread(b,r,r) | ratsimp

Contoh 6: Jarak Minimal pada Bidang

Catatan awal
Fungsi yang, ke titik M di bidang, menetapkan jarak AM antara titik tetap A dan M, memiliki garis level yang
agak sederhana: lingkaran berpusat di A.

>&remvalue();
>A=[-1,-1];
>function d1(x,y):=sqrt((x-A[1])^2+(y-A[2])^2)
>fcontour("d1",xmin=-2,xmax=0,ymin=-2,ymax=0,hue=1, ...
 title="If you see ellipses, please set your window square"):



dan grafiknya juga agak sederhana: bagian atas kerucut:

>plot3d("d1",xmin=-2,xmax=0,ymin=-2,ymax=0):

Tentu saja minimal 0 dicapai di A.

Dua titik

Sekarang kita melihat fungsi MA + MB dimana A dan B adalah dua titik (tetap). Ini adalah "fakta yang
terkenal" bahwa kurva level adalah elips, titik fokusnya adalah A dan B; kecuali untuk minimum AB yang
konstan pada segmen [AB]:

>B=[1,-1];
>function d2(x,y):=d1(x,y)+sqrt((x-B[1])^2+(y-B[2])^2)
>fcontour("d2",xmin=-2,xmax=2,ymin=-3,ymax=1,hue=1):



Grafiknya lebih menarik:

>plot3d("d2",xmin=-2,xmax=2,ymin=-3,ymax=1):

Batasan ke baris (AB) lebih terkenal:

>plot2d("abs(x+1)+abs(x-1)",xmin=-3,xmax=3):



Tiga titik

Sekarang hal-hal menjadi kurang sederhana: Sedikit kurang diketahui bahwa MA+MB+MC mencapai
minimumnya pada satu titik bidang tetapi untuk menentukannya kurang sederhana:

1) Jika salah satu sudut segitiga ABC lebih dari 120° (katakanlah dalam A), maka minimum tercapai pada
titik ini (katakanlah AB+AC).

Contoh:

>C=[-4,1];
>function d3(x,y):=d2(x,y)+sqrt((x-C[1])^2+(y-C[2])^2)
>plot3d("d3",xmin=-5,xmax=3,ymin=-4,ymax=4);
>insimg;

>fcontour("d3",xmin=-4,xmax=1,ymin=-2,ymax=2,hue=1,title="The minimum is on A");
>P=(A_B_C_A)'; plot2d(P[1],P[2],add=1,color=12);
>insimg;

2) Tetapi jika semua sudut segitiga ABC kurang dari 120°, minimum berada pada titik F di bagian dalam
segitiga, yang merupakan satu-satunya titik yang melihat sisi ABC dengan sudut yang sama (lalu masing-
masing 120°) :

>C=[-0.5,1];
>plot3d("d3",xmin=-2,xmax=2,ymin=-2,ymax=2):



>fcontour("d3",xmin=-2,xmax=2,ymin=-2,ymax=2,hue=1,title="The Fermat point");
>P=(A_B_C_A)'; plot2d(P[1],P[2],add=1,color=12);
>insimg;

Merupakan kegiatan yang menarik untuk mewujudkan gambar di atas dengan perangkat lunak geometri;
sebagai contoh, saya tahu soft tertulis di Java yang memiliki instruksi "garis kontur" ...

Semua ini di atas telah ditemukan oleh seorang hakim Prancis bernama Pierre de Fermat; dia menulis
surat kepada para penggila lainnya seperti pendeta Marin Mersenne dan Blaise Pascal yang bekerja di
bagian pajak penghasilan. Jadi titik unik F sehingga FA + FB + FC minimal disebut titik Fermat segitiga.
Tetapi tampaknya beberapa tahun sebelumnya, Torriccelli Italia telah menemukan titik ini sebelum Fermat
melakukannya! Pokoknya tradisinya adalah mencatat poin ini ...

Empat titik

Langkah selanjutnya adalah menambahkan titik D ke-4 dan mencoba meminimalkan MA + MB + MC +
MD; katakanlah bahwa Anda adalah operator TV kabel dan ingin mencari di bidang mana Anda harus
meletakkan antena sehingga Anda dapat memberi makan empat desa dan menggunakan kabel sesedikit
mungkin!

>D=[1,1];
>function d4(x,y):=d3(x,y)+sqrt((x-D[1])^2+(y-D[2])^2)
>plot3d("d4",xmin=-1.5,xmax=1.5,ymin=-1.5,ymax=1.5):



>fcontour("d4",xmin=-1.5,xmax=1.5,ymin=-1.5,ymax=1.5,hue=1);
>P=(A_B_C_D)'; plot2d(P[1],P[2],points=1,add=1,color=12);
>insimg;

Masih ada minimum dan tidak ada yang dicapai pada simpul A, B, C atau D:

>function f(x):=d4(x[1],x[2])
>neldermin("f",[0.2,0.2])

[0.142858,  0.142857]

Tampaknya dalam kasus ini, koordinat titik optimal rasional atau mendekati rasional ...

Sekarang ABCD adalah bujur sangkar, kami berharap bahwa titik optimal adalah pusat ABCD:

>C=[-1,1];
>plot3d("d4",xmin=-1,xmax=1,ymin=-1,ymax=1):



>fcontour("d4",xmin=-1.5,xmax=1.5,ymin=-1.5,ymax=1.5,hue=1);
>P=(A_B_C_D)'; plot2d(P[1],P[2],add=1,color=12,points=1);
>insimg;

Contoh 7: Bola Dandelin dengan Povray
Anda dapat menjalankan demonstrasi ini, jika Anda memiliki Povray diinstal, dan pvengine.exe di jalur
program.

Pertama kami menghitung jari-jari bola.

Jika Anda melihat gambar di bawah, Anda melihat bahwa kita membutuhkan dua lingkaran yang
menyentuh dua garis yang membentuk kerucut, dan satu garis yang membentuk bidang yang memotong
kerucut.

Kami menggunakan file geometry.e dari Euler untuk ini.

>load geometry;

Pertama, dua garis yang membentuk kerucut.

>g1 &= lineThrough([0,0],[1,a])

                             [- a, 1, 0]

>g2 &= lineThrough([0,0],[-1,a])

                            [- a, - 1, 0]

Lalu baris ketiga.

>g &= lineThrough([-1,0],[1,1])



                             [- 1, 2, 1]

Kita merencanakan semuanya sejauh ini.

>setPlotRange(-1,1,0,2);
>color(black); plotLine(g(),"")
>a:=2; color(blue); plotLine(g1(),""), plotLine(g2(),""):

Sekarang kita ambil titik umum pada sumbu y.

>P &= [0,u]

                                [0, u]

Hitung jarak ke g1.

>d1 &= distance(P,projectToLine(P,g1)); $d1

Hitung jarak ke g.

>d &= distance(P,projectToLine(P,g)); $d

Dan temukan pusat kedua lingkaran, di mana jaraknya sama.

>sol &= solve(d1^2=d^2,u); $sol

Ada dua solusi.

Kami mengevaluasi solusi simbolis, dan menemukan kedua pusat, dan kedua jarak.

>u := sol()

[0.333333,  1]

>dd := d()

[0.149071,  0.447214]

Plot lingkaran ke dalam gambar.

>color(red);
>plotCircle(circleWithCenter([0,u[1]],dd[1]),"");
>plotCircle(circleWithCenter([0,u[2]],dd[2]),"");
>insimg;



Plot dengan Povray
Selanjutnya kami merencanakan semuanya dengan Povray. Perhatikan bahwa Anda mengubah perintah
apa pun dalam urutan perintah Povray berikut, dan menjalankan kembali semua perintah dengan Shift-
Return.

Pertama kita memuat fungsi povray.

>load povray;
>defaultpovray="C:\Program Files\POV-Ray\v3.7\bin\pvengine.exe"

C:\Program Files\POV-Ray\v3.7\bin\pvengine.exe

Kita mengatur adegan dengan tepat.

>povstart(zoom=11,center=[0,0,0.5],height=10°,angle=140°);

Selanjutnya kita menulis dua bidang ke file Povray.

>writeln(povsphere([0,0,u[1]],dd[1],povlook(red)));
>writeln(povsphere([0,0,u[2]],dd[2],povlook(red)));

Dan kerucutnya, transparan.

>writeln(povcone([0,0,0],0,[0,0,a],1,povlook(lightgray,1)));

Kami menghasilkan pesawat terbatas pada kerucut.

>gp=g();
>pc=povcone([0,0,0],0,[0,0,a],1,"");
>vp=[gp[1],0,gp[2]]; dp=gp[3];
>writeln(povplane(vp,dp,povlook(blue,0.5),pc));

Sekarang kami menghasilkan dua titik pada lingkaran, di mana bola menyentuh kerucut.

>function turnz(v) := return [-v[2],v[1],v[3]]
>P1=projectToLine([0,u[1]],g1()); P1=turnz([P1[1],0,P1[2]]);
>writeln(povpoint(P1,povlook(yellow)));
>P2=projectToLine([0,u[2]],g1()); P2=turnz([P2[1],0,P2[2]]);
>writeln(povpoint(P2,povlook(yellow)));

Kemudian kami menghasilkan dua titik di mana bola menyentuh bidang. Ini adalah fokus elips.

>P3=projectToLine([0,u[1]],g()); P3=[P3[1],0,P3[2]];
>writeln(povpoint(P3,povlook(yellow)));
>P4=projectToLine([0,u[2]],g()); P4=[P4[1],0,P4[2]];
>writeln(povpoint(P4,povlook(yellow)));

Selanjutnya kita menghitung perpotongan P1P2 dengan bidang.

>t1=scalp(vp,P1)-dp; t2=scalp(vp,P2)-dp; P5=P1+t1/(t1-t2)*(P2-P1);
>writeln(povpoint(P5,povlook(yellow)));

Kami menghubungkan titik dengan segmen garis.

>writeln(povsegment(P1,P2,povlook(yellow)));
>writeln(povsegment(P5,P3,povlook(yellow)));
>writeln(povsegment(P5,P4,povlook(yellow)));

Sekarang kami membuat pita abu-abu, di mana bola menyentuh kerucut.



>pcw=povcone([0,0,0],0,[0,0,a],1.01);
>pc1=povcylinder([0,0,P1[3]-defaultpointsize/2],[0,0,P1[3]+defaultpointsize/2],1);
>writeln(povintersection([pcw,pc1],povlook(gray)));
>pc2=povcylinder([0,0,P2[3]-defaultpointsize/2],[0,0,P2[3]+defaultpointsize/2],1);
>writeln(povintersection([pcw,pc2],povlook(gray)));

Mulai program Povray.

>povend();

Untuk mendapatkan Anaglyph ini, kita perlu memasukkan semuanya ke dalam fungsi scene. Fungsi ini
akan digunakan dua kali nanti.

>function scene () ...
 global a,u,dd,g,g1,defaultpointsize;
 writeln(povsphere([0,0,u[1]],dd[1],povlook(red)));
 writeln(povsphere([0,0,u[2]],dd[2],povlook(red)));
 writeln(povcone([0,0,0],0,[0,0,a],1,povlook(lightgray,1)));
 gp=g();
 pc=povcone([0,0,0],0,[0,0,a],1,"");
 vp=[gp[1],0,gp[2]]; dp=gp[3];
 writeln(povplane(vp,dp,povlook(blue,0.5),pc));
 P1=projectToLine([0,u[1]],g1()); P1=turnz([P1[1],0,P1[2]]);
 writeln(povpoint(P1,povlook(yellow)));
 P2=projectToLine([0,u[2]],g1()); P2=turnz([P2[1],0,P2[2]]);
 writeln(povpoint(P2,povlook(yellow)));
 P3=projectToLine([0,u[1]],g()); P3=[P3[1],0,P3[2]];
 writeln(povpoint(P3,povlook(yellow)));
 P4=projectToLine([0,u[2]],g()); P4=[P4[1],0,P4[2]];
 writeln(povpoint(P4,povlook(yellow)));
 t1=scalp(vp,P1)-dp; t2=scalp(vp,P2)-dp; P5=P1+t1/(t1-t2)*(P2-P1);
 writeln(povpoint(P5,povlook(yellow)));
 writeln(povsegment(P1,P2,povlook(yellow)));
 writeln(povsegment(P5,P3,povlook(yellow)));
 writeln(povsegment(P5,P4,povlook(yellow)));
 pcw=povcone([0,0,0],0,[0,0,a],1.01);
 pc1=povcylinder([0,0,P1[3]-defaultpointsize/2],[0,0,P1[3]+defaultpointsize/2],1);
 writeln(povintersection([pcw,pc1],povlook(gray)));
 pc2=povcylinder([0,0,P2[3]-defaultpointsize/2],[0,0,P2[3]+defaultpointsize/2],1);
 writeln(povintersection([pcw,pc2],povlook(gray)));
 endfunction

Anda membutuhkan kacamata merah / cyan untuk mengapresiasi efek berikut.

>povanaglyph("scene",zoom=11,center=[0,0,0.5],height=10°,angle=140°);



Contoh 8: Geometri Bumi
Di notebook ini, kami ingin melakukan beberapa komputasi bola. Fungsi-fungsi tersebut terdapat dalam file
"spherical.e" di folder contoh. Kita perlu memuat file itu dulu.

>load spherical.e

Spherical functions for Euler. 

Untuk memasukkan posisi geografis, kami menggunakan vektor dengan dua koordinat dalam radian (utara
dan timur, nilai negatif untuk selatan dan barat). Berikut koordinat Kampus FMIPA UNY.

>FMIPA=[rad(-7,-46.467),rad(110,23.05)]

[-0.13569,  1.92657]

Anda dapat mencetak posisi ini dengan sposprint (cetak posisi bola).

>sposprint(FMIPA) // posisi garis lintang dan garis bujur FMIPA UNY

S 7°46.467' E 110°23.050'

Mari kita tambahkan dua kota lagi, Solo dan Semarang.

>Solo=[rad(-7,-34.333),rad(110,49.683)]; Semarang=[rad(-6,-59.05),rad(110,24.533)];
>sposprint(Solo), sposprint(Semarang),

S 7°34.333' E 110°49.683'
S 6°59.050' E 110°24.533'

Pertama kita menghitung vektor dari satu bola ke bola lainnya pada bola ideal. Vektor ini adalah [heading,
distance] dalam radian. Untuk menghitung jarak di bumi, kita mengalikan dengan jari-jari bumi pada garis
lintang 7°.

>br=svector(FMIPA,Solo); degprint(br[1]), br[2]*rearth(7°)->km // perkiraan jarak FMIPA-Solo

65°20'26.60''
53.8945384608

Ini adalah perkiraan yang bagus. Rutinitas berikut menggunakan perkiraan yang lebih baik. Pada jarak
yang begitu dekat hasilnya hampir sama.

>esdist(FMIPA,Semarang)->" km", // perkiraan jarak FMIPA-Semarang

88.0114026318 km

Ada fungsi untuk heading, dengan mempertimbangkan bentuk bumi yang elips. Sekali lagi, kami mencetak
dengan cara yang canggih.

>sdegprint(esdir(FMIPA,Solo))

     65.34°

Sudut segitiga melebihi 180° pada bola.



>asum=sangle(Solo,FMIPA,Semarang)+sangle(FMIPA,Solo,Semarang)+sangle(FMIPA,Semarang,Solo); degprint(asum)

180°0'10.77''

Ini dapat digunakan untuk menghitung luas segitiga. Catatan: Untuk segitiga kecil, ini tidak akurat karena
kesalahan pengurangan dalam asum-pi.

>(asum-pi)*rearth(48°)^2->" km^2", //perkiraan luas segitiga FMIPA-Solo-Semarang

2116.02948749 km^2

Ada fungsi untuk ini, yang menggunakan garis lintang rata-rata segitiga untuk menghitung jari-jari bumi,
dan menangani kesalahan pembulatan untuk segitiga yang sangat kecil.

>esarea(Solo,FMIPA,Semarang)->" km^2", //perkiraan yang sama dengan fungsi esarea()

2123.64310526 km^2

Kami juga dapat menambahkan vektor ke posisi. Vektor berisi heading dan jarak, keduanya dalam radian.
Untuk mendapatkan vektor, kami menggunakan svector. Untuk menambahkan vektor ke posisi, kami
menggunakan saddvector.

>v=svector(FMIPA,Solo); sposprint(saddvector(FMIPA,v)), sposprint(Solo),

S 7°34.333' E 110°49.683'
S 7°34.333' E 110°49.683'

Fungsi-fungsi ini mengasumsikan bola yang ideal. Hal yang sama di bumi.

>sposprint(esadd(FMIPA,esdir(FMIPA,Solo),esdist(FMIPA,Solo))), sposprint(Solo),

S 7°34.333' E 110°49.683'
S 7°34.333' E 110°49.683'

Mari kita beralih ke contoh yang lebih besar, Tugu Jogja dan Monas Jakarta (menggunakan Google Earth
untuk mencari koordinatnya).

>Tugu=[-7.7833°,110.3661°]; Monas=[-6.175°,106.811944°];
>sposprint(Tugu), sposprint(Monas)

S 7°46.998' E 110°21.966'
S 6°10.500' E 106°48.717'

Menurut Google Earth, jaraknya 429,66 km. Kami mendapatkan perkiraan yang bagus.

>esdist(Tugu,Monas)->" km", // perkiraan jarak Tugu Jogja - Monas Jakarta

431.565659488 km

Judulnya sama dengan yang dihitung di Google Earth.

>degprint(esdir(Tugu,Monas))

294°17'2.85''

Namun, kita tidak lagi mendapatkan posisi target yang tepat, jika kita menambahkan heading dan jarak ke
posisi semula. Hal ini terjadi, karena kita tidak menghitung fungsi invers secara tepat, tetapi mengambil
perkiraan jari-jari bumi di sepanjang jalan.

>sposprint(esadd(Tugu,esdir(Tugu,Monas),esdist(Tugu,Monas)))

S 6°10.500' E 106°48.717'

Namun, kesalahannya tidak besar.

>sposprint(Monas),

S 6°10.500' E 106°48.717'

Tentunya kita tidak bisa berlayar dengan tujuan yang sama dari satu tujuan ke tujuan lainnya, jika kita ingin
mengambil jalur terpendek. Bayangkan, Anda terbang NE mulai dari titik mana pun di bumi. Kemudian
Anda akan berputar ke kutub utara. Lingkaran besar tidak mengikuti arah yang konstan!

Perhitungan berikut menunjukkan bahwa kami jauh dari tujuan yang benar, jika kami menggunakan tajuk
yang sama selama perjalanan kami.

>dist=esdist(Tugu,Monas); hd=esdir(Tugu,Monas);

Sekarang kita tambahkan 10 kali sepersepuluh jaraknya, menggunakan heading ke Monas, kita sampai di
Tugu.



>p=Tugu; loop 1 to 10; p=esadd(p,hd,dist/10); end;

Hasilnya masih jauh.

>sposprint(p), skmprint(esdist(p,Monas))

S 6°11.250' E 106°48.372'
     1.529km

Sebagai contoh lain, mari kita ambil dua titik di bumi pada ketinggian yang sama.

> P1=[30°,10°]; P2=[30°,50°];

Jalur terpendek dari P1 ke P2 bukanlah lingkaran dengan garis lintang 30°, tetapi jalur yang lebih pendek
mulai 10° lebih jauh ke utara di P1.

>sdegprint(esdir(P1,P2))

     79.69°

Tapi, jika kita mengikuti pembacaan kompas ini, kita akan berputar ke kutub utara! Jadi kita harus
menyesuaikan arah tujuan kita di sepanjang jalan. Untuk tujuan kasar, kami menyesuaikannya pada 1/10
dari jarak total.

>p=P1;  dist=esdist(P1,P2); ...
   loop 1 to 10; dir=esdir(p,P2); sdegprint(dir), p=esadd(p,dir,dist/10); end;

     79.69°
     81.67°
     83.71°
     85.78°
     87.89°
     90.00°
     92.12°
     94.22°
     96.29°
     98.33°

Jaraknya tidak tepat, karena kita akan menambahkan sedikit kesalahan, jika kita mengikuti tajuk yang
sama terlalu lama.

>skmprint(esdist(p,P2))

     0.203km

Kami mendapatkan perkiraan yang baik, jika kami menyesuaikan heading setelah setiap 1/100 dari total
jarak dari Tugu ke Monas.

>p=Tugu; dist=esdist(Tugu,Monas); ...
   loop 1 to 100; p=esadd(p,esdir(p,Monas),dist/100); end;
>skmprint(esdist(p,Monas))

     0.000km

Untuk keperluan navigasi, kita bisa mendapatkan urutan posisi GPS di sepanjang lingkaran besar menuju
Monas dengan fungsi navigasi.

>load spherical; v=navigate(Tugu,Monas,10); ...
   loop 1 to rows(v); sposprint(v[#]), end;

S 7°46.998' E 110°21.966'
S 7°37.422' E 110°0.573'
S 7°27.829' E 109°39.196'
S 7°18.219' E 109°17.834'
S 7°8.592' E 108°56.488'
S 6°58.948' E 108°35.157'
S 6°49.289' E 108°13.841'
S 6°39.614' E 107°52.539'
S 6°29.924' E 107°31.251'
S 6°20.219' E 107°9.977'
S 6°10.500' E 106°48.717'

Kami menulis sebuah fungsi, yang menggambarkan bumi, dua posisi, dan posisi di antaranya.

>function testplot ...
 useglobal;
 plotearth;
 plotpos(Tugu,"Tugu Jogja"); plotpos(Monas,"Tugu Monas");
 plotposline(v);
 endfunction

Sekarang plot semuanya.

>plot3d("testplot",angle=25, height=6,>own,>user,zoom=4):



Atau gunakan plot3d untuk mendapatkan tampilan anaglyphnya. Ini terlihat sangat bagus dengan
kacamata merah / cyan.

>plot3d("testplot",angle=25,height=6,distance=5,own=1,anaglyph=1,zoom=4):

Latihan
1. Gambarlah segi-n beraturan jika diketahui titik pusat O, n, dan jarak titik pusat ke titik-titik sudut segi-n
tersebut (jari-jari lingkaran luar segi-n), r.

Petunjuk:

Besar sudut pusat yang menghadap masing-masing sisi segi-n adalah (360/n).
Titik-titik sudut segi-n merupakan perpotongan lingkaran luar segi-n dan garis-garis yang
melalui pusat dan saling membentuk sudut sebesar kelipatan (360/n).
Untuk n ganjil, pilih salah satu titik sudut adalah di atas.
Untuk n genap, pilih 2 titik di kanan dan kiri lurus dengan titik pusat.
Anda dapat menggambar segi-3, 4, 5, 6, 7, dst beraturan.

2. Gambarlah suatu parabola yang melalui 3 titik yang diketahui.

Petunjuk:
- Misalkan persamaan parabolanya y= ax^2+bx+c.
- Substitusikan koordinat titik-titik yang diketahui ke persamaan tersebut.
- Selesaikan SPL yang terbentuk untuk mendapatkan nilai-nilai a, b, c.

3. Gambarlah suatu segi-4 yang diketahui keempat titik sudutnya, misalnya A, B, C, D.
- Tentukan apakah segi-4 tersebut merupakan segi-4 garis singgung (sisinya-sisintya merupakan garis
singgung lingkaran yang sama yakni lingkaran dalam segi-4 tersebut).
- Suatu segi-4 merupakan segi-4 garis singgung apabila keempat garis bagi sudutnya bertemu di satu titik.
- Jika segi-4 tersebut merupakan segi-4 garis singgung, gambar lingkaran dalamnya.
- Tunjukkan bahwa syarat suatu segi-4 merupakan segi-4 garis singgung apabila hasil kali panjang sisi-sisi
yang berhadapan sama.

4. Gambarlah suatu ellips jika diketahui kedua titik fokusnya, misalnya P dan Q. Ingat ellips dengan fokus
P dan Q adalah tempat kedudukan titik-titik yang jumlah jarak ke P dan ke Q selalu sama (konstan).

5. Gambarlah suatu hiperbola jika diketahui kedua titik fokusnya, misalnya P dan Q. Ingat ellips dengan
fokus P dan Q adalah tempat kedudukan titik-titik yang selisih jarak ke P dan ke Q selalu sama (konstan).

Jawab:



1.

>o &:= [0,0]; z=circleWithCenter(o,5);
>color(1); setPlotRange(5); plotPoint(o); plotCircle(z);
>A=[-5,0]; plotPoint(A,"A");
>B=[5,0]; plotPoint(B,"B");
>plotSegment(A,B,""):

>z1=circleWithCenter(A,distance(A,o));
>z2=circleWithCenter(B,distance(B,o));
>f=circleCircleIntersections(z1,z);
>g=circleCircleIntersections(z,z2);
>h=circleCircleIntersections(z2,z);
>i=circleCircleIntersections(z,z1);
>r=lineThrough(f,h); s=lineThrough(g,i);
>setPlotRange(8); plotPoint(o); plotCircle(z); plotPoint(A,"A"); plotPoint(B,"B"); plotSegment(A,B,"");
>color(4); plotCircle(z1); plotCircle(z2); plotPoint(f); plotPoint(g); plotPoint(h); plotPoint(i);
>color(5); plotLine(r); plotLine(s):

>color(6); plotSegment(A,f,""); plotSegment(A,i,""); plotSegment(f,g,""); ...
  plotSegment(g,B,""); plotSegment(B,h,""); plotSegment(h,i,""):



2.

>setPlotRange(5); D=[1,0]; E=[4,0]; F=[0,-4];
>plotPoint(D,"D"); plotPoint(E,"E"); plotPoint(F,"F"):

>sol &= solve([a+b=-c,16*a+4*b=-c,c=-4],[a,b,c])

                     [[a = - 1, b = 5, c = - 4]]

>function y&=-x^2+5*x-4

                               2
                            - x  + 5 x - 4

>plot2d("-x^2+5*x-4",-5,5,-5,5):



3.

>setPlotRange(-5,5,-5,5);
>A=[-2,2]; plotPoint(A,"A");
>B=[2,2]; plotPoint(B,"B");
>C=[2,-2]; plotPoint(C,"C");
>D=[-2,-2]; plotPoint(D,"D");
>plotSegment(A,B);
>plotSegment(B,C);
>plotSegment(C,D);
>plotSegment(D,A);
>plotSegment(A,C,"k1");
>plotSegment(B,D,"k2");
>k1=lineThrough(A,C);
>k2=lineThrough(B,D);
>p=lineIntersection(k1,k2);
>plotLine(k1); plotLine(k2);
>plotPoint(p, "P");
>r=norm(p-projectToLine(p,lineThrough(A,B)))

2

>plotCircle(circleWithCenter(p,r),"lingkaran dalam segi-4 ABCD"):

>AB=norm(A-B) // panjang sisi AB

4

>CD=norm(C-D) // panjang sisi CD

4

>AD=norm(A-D) // panjang sisi AD



4

>BC=norm(B-C) // panjang sisi BC

4

>AB.CD

16

>AD.BC

16

4.

>P=[-1,-1]; Q=[1,-1];
>function d1(x,y):=sqrt((x-P[1])^2+(y-P[2])^2)
>Q=[1,-1]; function d2(x,y):=sqrt((x-P[1])^2+(y-P[2])^2)+sqrt((x-Q[1])^2+(y-Q[2])^2)
>fcontour("d2",xmin=-2,xmax=2,ymin=-3,ymax=1,hue=1):

>plot3d("d2",xmin=-2,xmax=2,ymin=-3,ymax=1):

5.

>P=[-1,-1]; Q=[1,-1];
>function d1(x,y):=sqrt((x-p[1])^2+(y-p[2])^2)
>Q=[1,-1]; function d2(x,y):=sqrt((x-P[1])^2+(y-P[2])^2)+sqrt((x+Q[1])^2+(y+Q[2])^2)
>fcontour("d2",xmin=-2,xmax=2,ymin=-3,ymax=1,hue=1):



>plot3d("d2",xmin=-2,xmax=2,ymin=-3,ymax=1):

>plot2d("abs(x+1)+abs(x-1)",xmin=-3,xmax=3):

>      


